
Suchen und Ersetzen in Dokumentkomponenten 
Volker Thormählen, 24. Mai 2017 

1 Aufgabe 
In diesem Beitrag wird beschrieben, wie bestimmte Komponenten eines WORD-Dokuments 
durchlaufen werden können, um dort mit der in WORD standardmäßig eingebauten Funktionalität 
„Suchen und Ersetzen“ die erforderliche(n) Korrektur(en) interaktiv durchzuführen. 
In einem Word-Dokument können bis zu 17 Komponenten vorkommen. Dazu zählt nicht nur der sog. 
Haupttext, sondern möglicherweise auch andere Dokumentkomponenten wie Fußnoten, Endnoten, 
Kopfzeilen, Fußzeilen usw. (vgl. Listing 3) 

2 Lösung 
Die Lösung beinhaltet 2 Subprozeduren und 1 Funktionsprozedur: 

• Die Subprozedur FindReplaceAnywhere entfernt 1 Leerzeichen vor Satzzeichen bzw. 
schließenden Klammern im aktuellen Dokument (s. Listing 1) 

• Die Subprozedur SearchAndReplaceInStory vollzieht die gerade genannte Aufgabe für die 
jeweilige Dokumentkomponente (s. Listing 2) 

o Codezeile 16 bewirkt die Markierung einer Fundstelle 
o Codezeile 19 ff beinhalten einen Benutzerdialog, wie bezüglich der Fundstelle 

entschieden und vorgegangen wird 

• Die Funktionsprozedur StoryTypeDescript ermittelt die jeweilige Bezeichnung einer 
Dokumentkomponente (s. Listing 3) 

Die Anwendung endet mit folgender Erfolgsmeldung, falls nur 1 Fundstelle ermittelt und ersetzt 
wurde: 

 

Abbildung 1: Erfolgsmeldung am Ende der Anwendung 

3 VBA1-Quellcode der Lösung 
 
Sub FindReplaceAnywhere() 1 
    ' Aufgabe: Leerzeichen entfernen vor Satzzeichen bzw. schließenden Klammern 2 
    '          in best. Dokumentkomponenten. 3 
    Const conTitle  As String = "In Dokumentkomponenten finden u. ersetzen." 4 
    Dim lngJunk     As Long     ' siehe weiter unten 5 
    Dim lngChrCnt1  As Long     ' Zeichenzahl im Dokument am Anfang 6 
    Dim lngChrDiff  As Long     ' Entfernte Zeichenzahl 7 
    Dim rngStory    As Range    ' Dokumentkomponente 8 
    Dim dtmStart    As Date     ' Startzeit der aktuellen Subprozedur 9 
    Dim strRunTm    As String   ' Laufzeit (in Stunden, Minuten, Sekunden) 10 
    Dim arrFnd      As Variant  ' Datenbereich für 9 Suchbegriffe 11 
    Dim arrRpl      As Variant  ' dito für zugehörige Ersetzungen 12 
    On Error GoTo ErrorHandler 13 
    dtmStart = Now 14 
    lngChrCnt1 = ActiveDocument.Characters.Count 15 
    arrFnd = Array(" ,", " ;", " .", " :", " !", " ?", " )", " ]", " }") 16 
    arrRpl = Array(",", ";", ".", ":", "!", "?", ")", "]", "}") 17 
    ' Behebung d. übersprungenen leeren Header-Footer-Problems 18 
    lngJunk = ActiveDocument.Sections(1).Headers(wdHeaderFooterPrimary).Range.StoryType 19 
    ' Best. Dokumentkomponenten durchlaufen 20 
    For Each rngStory In ActiveDocument.StoryRanges 21 
        Do Until rngStory Is Nothing 22 
            On Error Resume Next 23 
            If Len(rngStory.Text) > 0 Then 24 
                SearchAndReplaceInStory rngStory, arrFnd, arrRpl 25 
            End If 26 
            Select Case rngStory.StoryType 27 
                Case 6, 7, 8, 9, 10, 11 28 
                    If Len(rngStory.Text) > 0 Then 29 
                        SearchAndReplaceInStory rngStory, arrFnd, arrRpl 30 
                    End If 31 
                Case Else 32 
                    ' nichts tun! 33 
            End Select 34 
            ' Die nächste verknüpfte Dokumentkomponente ermitteln. 35 
            On Error GoTo 0 36 
            Set rngStory = rngStory.NextStoryRange 37 
        Loop 38 
    Next rngStory 39 
    ' Laufzeit ermitteln 40 
    strRunTm = Format(Now - dtmStart, "hh:mm:ss") 41 
    ' Entfernte Zeichenzahl ermitteln 42 
    lngChrDiff = lngChrCnt1 - ActiveDocument.Characters.Count 43 
    MsgBox "Fertig: " & CStr(lngChrDiff) & " Leerzeichen im aktuellen Dokument entfernt." & _ 44 
           vbCrLf & "Laufzeit: " & strRunTm & " hh:m:ss", vbInformation, conTitle 45 
ExitPoint: 46 
    On Error Resume Next 47 
    Exit Sub 48 
ErrorHandler: 49 
    MsgBox "Laufzeitfehler: " & Err.Number & ", " & Err.Description, vbCritical, conTitle 50 
    Resume ExitPoint 51 
End Sub 52 

Listing 1: VBA-Quellcode der Subprozedur ‘FindReplaceAnywhere’ 

 

 
1 Das Kürzel VBA steht für Visual Basic for Applications. Das ist eine proprietäre Skriptsprache der Firma 

Microsoft für die  sog. Office-Suite. 

Sub SearchAndReplaceInStory(ByVal rngStory As Range, _ 1 
                            ByVal arrFnd   As Variant, _ 2 
                            ByVal arrRpl   As Variant) 3 
    Dim intChanges  As Integer  ' Zähler f. Änderungen 4 
    Dim intHits     As Integer  ' Trefferzähler 5 
    Dim intLoop     As Integer  ' Scheifenzähler 6 
    Dim lngRslt     As Long     ' Rückgabewert der MsgBox (Benutzerdialog) 7 
    With rngStory.Find 8 
        For intLoop = 0 To UBound(arrFnd) 9 
            .ClearFormatting 10 
            .Text = arrFnd(intLoop) 11 
            .Replacement.Text = "" 12 
            .Execute 13 
            Do While .Found 14 
                intHits = intHits + 1 15 
                rngStory.Duplicate.Select   ' Aktuelle Fundstelle markieren. 16 
                ' Fensterinhalt verschieben, bis die Markierung darin angezeigt wird. 17 
                ActiveWindow.ScrollIntoView obj:=Selection.Range, Start:=True 18 
                lngRslt = MsgBox("Ersetzen:" & vbCr & vbTab & arrFnd(intLoop) & vbCr & _ 19 
                                    "durch:" & vbCr & vbTab & arrRpl(intLoop), vbYesNoCancel, _ 20 
                                    "In Dokumentkomponenten finden und ersetzen.") 21 
                Select Case lngRslt 22 
                    Case vbCancel 23 
                        Exit Sub 24 
                    Case vbYes 25 
                        rngStory.Text = arrRpl(intLoop) 26 
                        intChanges = intChanges + 1 27 
                    Case Else 28 
                        ' nichts tun! 29 
                End Select 30 
                rngStory.Collapse Direction:=wdCollapseEnd 31 
                .Execute 32 
            Loop 33 
        Next intLoop 34 
        If intHits > 0 Then 35 
            MsgBox CStr(intHits) & " Treffer in '" & StoryTypeDescript(rngStory.StoryType) & _ 36 
                "' mit " & CStr(intChanges) & " Änderung(en).", vbInformation, _ 37 
                "In Dokumentkomponenten finden und ersetzen." 38 
        End If 39 
    End With 40 
End Sub 41 

Listing 2: VBA-Quellcode der Subprozedur ‘SearchAndReplaceInStory’ 

Function StoryTypeDescript(ByVal intStoryType As Integer) As String 1 
    ' Aufgabe: Bezeichnung der jeweiligen Dokumentkomponente zurückgeben.2 2 
    Dim varStory(17) As Variant 3 
    varStory(1) = "Haupttext" 4 
    varStory(2) = "Fußnote" 5 
    varStory(3) = "Endnote" 6 
    varStory(4) = "Kommentar" 7 
    varStory(5) = "Textrahmen" 8 
    varStory(6) = "Gerade Kopfzeile" 9 
    varStory(7) = "Primäre Kopfzeile" 10 
    varStory(8) = "Gerade Fußzeile" 11 
    varStory(9) = "Primäre Fußzeile" 12 
    varStory(10) = "Erste Kopfzeile" 13 
    varStory(11) = "Erste Fußzeile" 14 
    varStory(12) = "Fußnotentrennlinie" 15 
    varStory(13) = "Fußnoten-Fortsetzungstrennlinie" 16 
    varStory(14) = "Fußnoten-Fortsetzungshinweis" 17 
    varStory(15) = "Endnotentrennlinie" 18 
    varStory(16) = "Endnoten-Fortsetzungstrennlinie" 19 
    varStory(17) = "Endnoten-Fortsetzungshinweis" 20 
    If intStoryType < 1 Or intStoryType > UBound(varStory) Then 21 
        StoryTypeDescript = vbNullString 22 
    Else 23 
        StoryTypeDescript = varStory(intStoryType) 24 
    End If 25 
End Function 26 

Listing 3: VBA-Quellcode der Funktionsprozedur ‘StoryTypeDescript‘ 

 
2 Quelle: Vgl. Document.StoryRanges-Eigenschaft (Word), URL: https://docs.microsoft.com/de-
de/office/vba/api/word.document.storyranges, gefunden am 18. Mai 2021 

https://docs.microsoft.com/de-de/office/vba/api/word.document.storyranges
https://docs.microsoft.com/de-de/office/vba/api/word.document.storyranges

Tabelle 1: Dokumentkomponenten (StoryTypes) in WORD3 

 

3 Quelle: René Probst, Tabelle der «StoryTypes» in Word, URL: https://www.kastenmaier.de/wp-

content/uploads/2007/07/Word_FAQ.pdf, gefunden am 24. Mai 2021 
 

https://www.kastenmaier.de/wp-content/uploads/2007/07/Word_FAQ.pdf
https://www.kastenmaier.de/wp-content/uploads/2007/07/Word_FAQ.pdf

